
ABSTRACT

In studying light and image transfer in sea water the influence of Fresnel surface
reflection is as significant as scattering and absorption phenomena. In these cases a
knowledge of the reflective properties of sea surface at different wind speeds is very
important. At present, little is published about these properties. We present here results
of numerical modeling of Fresnel light reflection coefficient of sea water as a function of
solar zenith angle and wind speed. The ray-tracing computer model was developed to
generate wave slopes and elevations. In order to generate a realistic sea surface the model
used Paul Hwang wave height spectrums. The final result of this paper is a simple
equation and very fast FORTRAN code to calculate Fresnel reflection coefficient of wavy
water surface and specular part of remote sensing reflectance.

INTRODUCTION

The optical remote sensing reflectance of the ocean consists of two parts: one that is connected
to the specular reflection of light from wavy water surface, and the second one that is due to the
diffuse light ascending from the water. The information about the optical state of water body is
contained in the internal light coming from the sea. In order to precisely estimate the second
internal component of the remote sensing reflection we need an approach to precisely estimate the
first, specular reflected component of light.

This presentation explains a new and very fast algorithm to compute specular reflection of solar
light from the wavy water surface. The specular light reflected from the water surface consists of two
parts. The first part is due to the Fresnel reflection from a wind-roughened surface, and the second
one is due to the reflection by whitecaps. The Fresnel part of reflection was computed using a Monte
Carlo method. The roughness of water surface was realistically modeled with the use of surface wave
energy distributions by Paul Hwang. The resulting reflection coefficient of light by a water surface
was obtained by averaging over billion realizations of reflection coefficient.  Results of these
calculations are expressed as a very precise regression that connects Fresnel reflection coefficient to
the solar angle and wind speed.

The proposed algorithm is implemented as a very fast FORTRAN code (See Appendices A and B)
to compute a specular reflection component of the remote sensing reflectance.

_________________________
Presented to the Seventh International Conference: Remote Sensing for Marine and Coastal
Environments, Miami, Florida, USA, 20-22 May 2002.



Proceedings of the Seventh International Conference on Remote Sensing for Marine and Coastal Environments
20-22 May, 2002, Miami, Florida, USA, ISSN 1066-3711, Published by: Veridan, P.O. Box 134008, Ann Arbor, MI 48113-4008, USA, 2002

00006 - 2

SPECULAR PART OF REMOTE SENSING REFLECTANCE

The specular part of remote sensing reflectance consists of two parts, the first one,
connected with Fresnel reflection coefficient of light from rough (wavy) water surface rF , and the
second one, related to reflection of light from sea foam or whitecaps rSF . It can be expressed as a
following expression:

r r rS F SF= + , (1)

The reflection by sea foam can be regarded as Lambertian and is expressed as follows:

r f u ASF F= ( ) p , (2)

where AF  is an albedo of the foam ( AF @ 0 6. ), and f u( ) is a portion of the sea surface covered with
whitecaps (foam), u  is a wind speed. The function f u( ) is defined by the following regression
(Frouin, 1996):

The Fresnel portion of the remote sensing reflectance is defined by the following equation:

r f u R u ZF F S= -[ ]1 ( ) ( , ) , (4)

where ZS  is an angle of incidence equal to a solar zenith angle for flat water surface. When wind
speed is equal to zero, there are no foam on a sea surface ( f u( ) = 0) and foam-related part of remote

sensing reflectance is equal to zero rSF = 0 , the surface is flat and r R ZF F S= 0 ( ), where R ZF S
0 ( ) is

Fresnel specular reflection coefficient from flat water surface, defined by the following equation:

(5)

here nw ª 1 341.  is a refractive index of water.

When wind started to blow, it roughens the water surface and the situation becomes more
complex: we need to average expression (5) over incidence angles. There are different approaches to
accomplish this procedure: 1) we can use a Cox and Munk (1954) sea water slope distribution, or 2)
we can use Monte Carlo method to generate wave slopes using wave energy spectrums taken from
experimental measurements (Haltrin, McBride III, and Weidemann, 2000). The second approach is
more complex and time-consuming, but it gives better results, especially for remote sensing
applications. To compute specular part of remote sensing reflectance in this paper we used the
second approach.

COMPUTATIONS OF FRESNEL REFLECTION COEFFICIENT

To generate sea water surface slopes we used a Monte Carlo program described in Haltrin,
McBride III, and Arnone (2001). The values of sea surface orientation allow us to compute light
reflection coefficient in each pixel of the sample surface. Averaging over M M¥  spatial and L
temporal realizations gives as an average Fresnel reflection coefficient for each value of wind speed
u  and zenith angle ZS . To compute actual values of Fresnel reflection coefficient we used the Paul
Hwang (1997) spectrum. This spectrum is specifically tailored for remote sensing problems to
produce correct values of mean square slopes of ocean waves.

The generated realizations of Fresnel reflection coefficients have been averaged over
100 100¥  pixels of sample sea surface areas and 80 time realizations to produce resulting angular
distributions of Fresnel reflection coefficient. The values of elevations and orientations in each pixel
have been obtained using one million computations, 1000 realizations for flat sine waves and 1000
significant points of wave number k  in energy spectrum range. To generate random numbers we used
Mersenne Twister random number generator (Matsumoto and Kurita, 1992, 1994) capable to
produce evenly distributed random numbers in a cube of 626 dimensions. So each value of Fresnel
reflection coefficient for any value of wind speed and zenith angle represents an average value of 800
billion individual computations.

FRESNEL REFLECTION COEFFICIENT BY WAVY WATER SURFACE

The results of extensive Monte Carlo computations are compactly represented as a following
regression equation (Haltrin, McBride III, and Arnone, 2001):

R u Z a u R Z a u R Z a u a u R ZF S F S F S F S( , ) ( ) ( ) ( ) ( ) ( ) ( ) ( )= + + +[ ]{ }0
0

1
0

2 3
0 , (6)

where R ZF S
0 ( ) is a Fresnel reflection coefficient of flat water surface given by Eq. (5), and wind-

speed-dependent coefficients a ui( ) are given by the following equations:

a u u u0
2( ) =  0.001 6.944831-1.912076 +0.03654833( ) ,    r2 =  0.9997, (7)

a u u u1
2( ) =  0.7431368 +0.0679787 - 0.0007171 ,    r2 =  0.9996, (8)

a u u u u2
2 30 00( ) =  0.5650262 +0.0061502 - 0.0239810 10695+ . ,   r2 =  0.9995, (9)

a u u u u3
2 30 00( ) =  - 0.4128083 - 0.1271037 +0.0283907 11706- . ,   r2 =  0.9991. (10)

Equations (1)-(6) represent a complete algorithm to compute the specular part of remote sensing
reflectance. This algorithm is programmed in FORTRAN and presented in Appendices A and B.

APPENDIX A: PROGRAM RSSPEC.F

! **********************************************************************
!                         program rsspec
! **********************************************************************
!           Vladimir I. Haltrin <haltrin@nrlssc.navy.mil>
!             last modification: February 20, 2002
! **********************************************************************
!  vwk         = wind speed over the sea surface in knots
!  vwm         = wind speed over the sea surface in m/sec
!  f, ff       = fraction of sea surface covered by foam
!  Zs(1:nSun)  = a massive of solar zenith angles
!  nWat        = refractive index of sea water
! **********************************************************************
      implicit   none
      integer    nSun,Ns,Nw,Nz,j,k
      parameter  (Nw=20,Nz=90)
      real       nWat,R,Afoam,Pi,rd,ws,dt,ff,phi,fr0,rsa,f(Nw)
      real       windspeed(Nw),vwk(Nw),vwm(Nw),Zs(Nz),RfrFl(Nz),Fresnel
      real       RfrWn(Nw,Nz),Rrs(Nw,Nz),Ffoam,FresRwind
      logical    knots

      open(11,file='rsspec.in',status='old')
         read(11,*) Afoam
         read(11,*) nWat
         read(11,*) knots
         read(11,*) Ns
         read(11,*) (windspeed(j),j=1,Ns)
         read(11,*) nSun
         read(11,*) (Zs(k), k=1, nSun) ! solar zenith angles in degrees
      close(11)

      Pi = 4.*ATAN(1.)
      rd = Pi/180.

      if (knots) then
         do j=1,Ns
            vwm(j) = 0.515*windspeed(j)
            vwk(j) = windspeed(j)
         end do
      else
         do j=1,Ns
            vwm(j) = windspeed(j)
            vwk(j) = windspeed(j)/0.515
         end do
      end if

      do k = 1, nSun                 ! solar zenith angle loop
         phi = rd*Zs(k)

 fr0 = Fresnel(nWat,phi) ! Fresnel reflection coefficient
         RfrFl(k) = fr0
         do j=1, Ns                 ! windspeed loop
            ws = vwm(j)
            ff = Ffoam(ws)
            f(j) = ff
            R = FresRwind(fr0,ws)
            RfrWn(j,k) = R
            Rrs(j,k) = ff*Afoam/Pi + (1.-ff)*R

         end do
      end do

      call Rout('RfrWnd.out',Afoam,nWat,Nw,Nz,Ns,nSun
     &                                 ,f,vwk,vwm,Zs,RfrFl,RfrWn)
      call Rout('RSspec.out',Afoam,nWat,Nw,Nz,Ns,nSun
     &                                 ,f,vwk,vwm,Zs,RfrFl,Rrs)

      end

! **********************************************************************
      real   function Fresnel(nWat,phi)
! ----------------------------------------------------------------------
!   phi    = angle (in radians) at which light is incident on sea
!   aRef   = angle of refraction (in radians)
!   Rpar   = Fresnel refrlection coefficient for parallel polarization
!   Rper   = Fresnel refrlection coefficient for perpend. polarization
! **********************************************************************
      implicit   none
      real       nWat,phi, aRef,aDif,aSum,Rpar,Rper

      if (phi .ne. 0.) then
         aRef = ASIN(SIN(phi)/nWat)
         aDif = phi-aRef
         aSum = phi+aRef
         Rpar = TAN(aDif)/TAN(aSum)
         Rper = SIN(aDif)/SIN(aSum)
         Fresnel = 0.5*(Rpar*Rpar+Rper*Rper)
      else
         aSum = (nWat-1.)/(nWat+1.)
         Fresnel = aSum*aSum
      end if

      return
      end

! **********************************************************************
      real function FresRwind(fr0,ws)
! ----------------------------------------------------------------------
!   Calculates Fresnel reflection coefficient of wavy water surface
!   for the case of Paul Hwang wave energy spectrum distribution
!   fr0  = Fresnel reflection coefficient of a flat water surface.
!   ws   = windspeed in m/sec, 0 <= ws <= 12 m/s.
!   See: V. I. Haltrin, W. E. McBride III, and R. A. Arnone, “Spectral
!   approach to calculate specular reflection of light from wavy water
!   surface,” - pp. 133-138 in Proceedings of D. S. Rozhdestvensky
!   Optical Society: International Conference Current Problems in Optics
!   of Natural Waters (ONW’2001), St. Petersburg, Russia, 2001.
! **********************************************************************
      implicit   none
      real       fr0,ws, a0,a1,a2,a3

       a0 = 0.001*(6.944831+ws*(-1.912076+0.03654833*ws))
       a1 = 0.7431368+ws*(0.0679787-0.0007171*ws)
       a2 = 0.5650262+ws*(0.0061502+ws*(-0.023981+0.0010695*ws))
       a3 = -0.4128083+ws*(-0.1271037+ws*(0.0283907-0.0011706*ws))
      FresRwind = a0+fr0*(a1+fr0*(a2+a3*fr0))

      return

      end

! **********************************************************************
      real   function Ffoam(ws)
! ----------------------------------------------------------------------
!     Calculates a fraction of water surface covered by foam
!     ws = windspeed in m/sec, 0 <= ws <= 12 m/s.
!     R. Frouin, Ocean Optics XIII, Halifax, Canada, unpublished.
! **********************************************************************
      implicit   none
      real       ws,f

      f = ws*ws*ws
      f = 1.2E-5*f*(ws**0.3)
      if (ws .gt. 9.) f = f*(0.225*ws-0.99)
      if (f  .gt. 1.) f = 1.
      Ffoam = f

      return
      end

! **********************************************************************
      subroutine Rout(fname,Afoam,nWat,Nw,Nz,Ns,nSun
     &                                      ,f,vwk,vwm,Zs,RfrFl,Rf)
! **********************************************************************
      implicit   none
      integer    Nw,Nz,Ns,nSun,k,j
      real       Afoam,nWat,f(Nw),vwk(Nw),vwm(Nw)
      real       Zs(Nz),RfrFl(Nz),Rf(Nw,Nz)
      character  tb,fname*10

      tb = CHAR(9)
      open(21, file=fname, status='new',recl=250)
         write(21,44) Afoam
         write(21,55) nWat
         write(21,66) '     f:',tb,0.,(tb,  f(j),j=1,Ns)
         write(21,66) 'Zs\vwk:',tb,0.,(tb,vwk(j),j=1,Ns)
         write(21,66) '-------------------------------------------------
     &----------------'
         write(21,66) 'Zs\vwm:',tb,0.,(tb,vwm(j),j=1,Ns)
         do k = 1, nSun
            write(21,77) Zs(k),tb,RfrFl(k),(tb,Rf(j,k),j=1,Ns)
         end do
      close(21)

   44 format(x,'Foam Albedo                    = ',f7.5)
   55 format(x,'Water refraction coefficient   = ',f7.5)
   66 format(a7,  21(a1,f10.5))
   77 format(f6.3,21(a1,f10.7))

      return
      end
! **********************************************************************

APPENDIX B: INPUT FILE RSSPEC.IN

 0.60 <-- Afoam, foam albedo (0.60)
 1.341 <-- nWat, water refractive index
 .false. <-- knots (if .true., windspeed is in knots, else -> in m/sec)
 6 <-- Ns;  windspeed(1:Ns):
 2. 4.00 6. 8. 10.0 12.
 22 <-- nSun;    sun zenith angles in degrees, ang(1:nSun):
 0.0 05. 10. 15 20. 25. 30. 35. 40. 45.
 50. 55. 60. 65. 70. 75. 77.5 80. 82.5 85. 87.5 89.9

